# Maintainer: Sergej Pupykin <arch+pub@sergej.pp.ru>
# Maintainer: Christian Babeux <christian.babeux@0x80.ca>
# Contributor: Yggdrasil <tetzank at web dot de>

pkgname=liburcu
pkgver=0.14.0
pkgrel=1
pkgdesc="LGPLv2.1 userspace RCU (read-copy-update) library"
arch=('loong64' 'x86_64')
url="https://lttng.org/urcu"
license=('LGPL2.1')
depends=('glibc')
source=(https://lttng.org/files/urcu/userspace-rcu-${pkgver}.tar.bz2{,.asc}
        userspace-rcu-loongarch64.patch)
sha256sums=('ca43bf261d4d392cff20dfae440836603bf009fce24fdc9b2697d837a2239d4f'
            'SKIP')
validpgpkeys=('2A0B4ED915F2D3FA45F5B16217280A9781186ACF')

prepare() {
    cd "$srcdir"/userspace-rcu-${pkgver}
    patch -p1 -i "$srcdir/userspace-rcu-loongarch64.patch"
    autoreconf -vfi
}

build() {
    cd "$srcdir"/userspace-rcu-${pkgver}
    ./configure --prefix=/usr
    make
}

package() {
    cd "$srcdir"/userspace-rcu-${pkgver}
    make install DESTDIR="$pkgdir"
}
